AD-A170 051 SOLUTION OF POISSON'S EQUATION USING METHOD OF MOMENTS: APPLICATION TO MOS DEVICES(U) RIT RESEARCH CORP ROCHESTER NY E ARVAS ET AL MAR 86 N00014-84-K-0532 1/1

UNCLASSIFIED F/G 20/12 NL



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

AD-A170 051

(12)

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER RITRC 001	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) SOLUTION OF POISSON'S EQUATION USING METHOD OF MOMENTS: APPLICATION TO MOS DEVICES		5. TYPE OF REPORT & PERIOD COVERED Technical Report # 01 Oct. 85 to 31 Mar 86
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Ercument Arvas, Perambur S. Neelakantaswamy, and Ibrahim R. Turkman		8. CONTRACT OR GRANT NUMBER(s) N00014-84-K-0532
9. PERFORMING ORGANIZATION NAME AND ADDRESS RIT Research Corporation 75 Highpower Road Rochester, NY 14623-3435		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS NR 613-005
11. CONTROLLING OFFICE NAME AND ADDRESS Department of Navy Office of Naval Research Arlington, VA 22217		12. REPORT DATE March 1986
		13. NUMBER OF PAGES 33
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

This document has been approved for public release and sale; its distribution is unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

JUL 0 7 1986

D

E

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

MOS DEVICES, POISSON'S EQUATION (TWO-DIMENSIONAL), METHOD OF MOMENTS,

THRESHOLD VOLTAGE, SURFACE POTENTIAL

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

An algorithm for the computation of solution to Poisson's equation in a two-dimensional domain is developed in terms of equivalent sources on the boundary. The region considered can be of arbitrary shape, and the boundary conditions can be Dirichlet, Neumann or mixed type. The solution is obtained by method of moments. Pulse expansion and point matching techniques are used. Computed results closely agree with the available data concerning MOS devices.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

S/N 0102-LF-014-6601

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

(12)

SOLUTION OF POISSON'S EQUATION USING METHOD OF MOMENTS: APPLICATION TO MOS DEVICES

by

Ercument Arvas
Electrical Engineering Department
Rochester Institute of Technology
One Lomb Memorial Drive
Rochester, New York 14623

Perambur S. Neelakantaswamy
RIT Research Corporation
75 Highpower Road
Rochester, New York 14623

and

Ibrahim R. Turkman
Electrical Engineering Department
Rochester Institute of Technology
Rochester, New York 14623

RIT Research Corporation
75 Highpower Road
Rochester, New York 14623

X

ltr per

A-1

March, 1986

86 5 9 027

CONTENTS

ABSTRACT

An algorithm for the computation of solution to Poisson's equation in a two-dimensional domain is developed in terms of equivalent sources on the boundary. The region considered can be of arbitrary shape, and the boundary conditions can be Dirichlet, Neumann or mixed type. The solution is obtained by method of moments. Pulse expansion and point matching techniques are used. Computed results closely agree with the available data concerning MOS devices.

I. INTRODUCTION

Poisson's equation is one of the most important differential equations of physics. For example, it can be used to find the threshold voltages of MOSFET's. When the channel length is small, the depletion-layer widths of the source and drain junctions are comparable to the channel length, and the potential distribution is two dimensional amenable for solution via Poisson's equation.

In this work we give a simple method for solving two-dimensional Poisson's equation in a region subject to general boundary conditions on the bounding surface. Equivalent surface charges are placed just outside the boundary and the total potential (produced by the impressed volume charges and the equivalent surface charges) is enforced to satisfy the boundary conditions. This transforms the boundary value problem into an integral equation for the equivalent surface charges. Then the method of moments [1] is used to solve the integral equation numerically.

II. STATEMENT OF PROBLEM

Consider a 2-dimensional region R bounded by the contour C as shown in Figure 1. The problem is to find the total potential ψ (x,y) in R which satisfies the Poisson's equation

$$\frac{\partial^2 \psi}{\partial x^2} + \frac{\partial^2 \psi}{\partial y^2} = -\rho_v/\varepsilon \qquad (1)$$

in R, with the boundary condition(s)

$$\alpha\psi + \beta \frac{\partial \psi}{\partial n} = \gamma \qquad (2)$$

on C.

In eqn. (1), ρ_v denote the volume charge density, and ε is the permittivity of the material in R.

In eqn. (2), α , β and γ denotes known functions defined on C.

Note that the general condition of eqn. (2) includes, as special cases, of Dirichlet (α =1, β =o) and Neumann (α =o, β =1) conditions.

The Laplace's equation is the special case of eqn. (1) with ρ_v = o The solution of Laplace's equation is given in detail in [2]. The present work is an extension of the work in [2], modified for Poisson's equation applied to MOS structures.

III. METHOD OF SOLUTION

In solving eqn. (1) subject to boundary condition of eqn. (2) we let

$$\psi = \phi_p + \phi_h \qquad (3)$$

where (4)

$$\nabla^2 \phi_p = \frac{-\rho_v}{\varepsilon} \quad \text{(in R)}$$

and (5)

$$\nabla^2 \phi_h = 0 \quad \text{(in R)}$$

The solution to eqn. (4) is

$$\phi_p = \frac{1}{2\pi\varepsilon} \iint_R \rho(\underline{r}') \ln \frac{k}{\underline{r} - \underline{r}'} \, dx'dy' \tag{6}$$

where $\underline{r}$ and $\underline{r}'$ denote the radius vector to the field and a source point respectively, $\rho(\underline{r}')$ is the value of the impressed charged density at $\underline{r}'$s and k is an arbitrary constant (taken as 100.0 in this work).

The Laplacian potential ϕ_h can be assumed to be produced by some equivalent surface charges, σ , outside R (Fig. 2). Hence ϕ_h is the solution of eqn. (5) subject to boundary condition

$$\alpha\phi_h + \beta \frac{\partial\phi_h}{\partial n} = \gamma - \alpha\phi_p - \beta \frac{\partial\phi_p}{\partial n} \quad \text{on } C \tag{7}$$

Since ϕ_h has the form

$$\phi_h = \frac{1}{2\pi\varepsilon} \iint_C \sigma \ln \frac{k}{\underline{r} - \underline{r}'} \, dx'dy' \tag{8}$$

we see that eqn. (7) is an integral equation for σ .

Note that (5) subject to the boundary condition of eqn. (7) is the same boundary value problem as the one considered in [2]. We use pulse expansion and point-matching techniques to solve this problem.

The approach involved is to first model the surface C by N planar strips and the assume a constant charge density on each segment. Satisfying the boundary condition of eqn. (7) at the center of each of N strips, gives N algebraic equations. The solution of these equations gives the value of the constant charge density on each strip. The details are elaborated in [2]. Once eqn. (7) is solved for σ , we obtain the total potential ψ using eqns. (8), (6) and (3).

IV. SAMPLE RESULTS

A FORTRAN program is written to implement the theory developed above.

The first test problem that we tried is shown in Fig. 3, where a line charge of $\rho_1 = 8.854\times10^{-12}$ C/m is placed at the center of a grounded rectangular boundary. The total potential was evaluated at the points A, B, C and D as shown.

Table 1 illustrates the convergence of the computed results as the number N of segments is increased. The exact result [3, eqn. 4-7.23] is also shown for comparison. The last column of the table shows the CPU time on a VAX 11/782.

A second test problem formulated to study a short channel MOSFET is described hereunder.

VI. APPLICATION TO MOS STRUCTURES

To demonstrate the applicability of the proposed numerical method to MOS structures, a test N-channel MOSFET illustrated in Fig. 4 is considered. The rectangular depletion region under the gate and its expanded view with the relevant boundary conditions are depicted in Fig. 5.

The notations followed are those detailed in [4]. Figs. 6 and 7 illustrate the surface potential Ψ (x, y = d) variation along the channel for 2 typical devices with channel lengths L = 1 μ m and 5 μ m respectively.

The corresponding threshold voltage ($V_T - V_{FB}$) versus channel length for drain voltage (V_D) of 0 and 5V is presented in Fig. 8.

For comparison, along with the computed data, the results obtained by (approximate) closed-form solution due to Poole and Kwong [4] are also shown in Figs. 6, 7 and 8.

Referring to these figures (Figs. 6, 7 and 8), close agreement between the results may be observed. Any deviation can be attributed to the approximations involved in the truncation of the series solution given in [4] and due to the variations in the values of d and V_{gm} considered in the analysis.

However, the present work indicates the applicability of the method of solution envisaged to the MOS structures. This method can be extended to a more realistic model of the MOS structure involving curved depletion boundaries and the depletion width (d) varying along the channel length.

Further, this steady state solution can be extended to study transient causes pertaining to ESD/EOS induced effects.

V. REFERENCES

[1] R. F. Harrington, Field Computations by Moment Methods. New York: Macmillan, 1968; reprinted by Krieger, Melbourne, FL, 1982.

[2] R. F. Harrington, K. Pontoppidan, P. Abrahamsen and N. C. Albertsen, "Computation of Lapacian Potentials by an equivalent-source method", Proc. IEE, Vol. 116, No. 10, pp. 1715-1720, Oct. 1969.

[3] S. Seely and A. D. Poularikas, Electromagnetics, Marcel Dekker, Inc., New York, 1979.

[4] D. R. Poole and D. L. Kwong, "Two Dimensional Analytical Modeling of Threshold Voltages of Short-Channel MOSFET's," IEEE Electron Device Letters, Vol. Edl-5, No. 11, pp. 443-448, Nov. 1984.

Table I

Potential computed for the problem of Fig. 3

N	A (0.25,0.75)	B (0.25,0.50)	C (0.1,0.25)	D (0.45,0.95)	CPU Time (Sec.)
4	0.05924	0.1133	0.01411	0.3145	3.28
8	0.07271	0.1268	0.02757	0.3280	3.55
12	0.07113	0.1226	0.02875	0.3244	4.03
16	0.07050	0.1221	0.02747	0.3238	4.54
20	0.07032	0.1219	0.02721	0.0236	5.15
24	0.07024	0.1218	0.02719	0.3235	5.92
32	0.07018	0.1217	0.02714	0.3234	7.86
40	0.07016	0.1217	0.02712	0.3234	10.69
60	0.07014	0.1216	0.02711	0.3234	21.43
80	0.07014	0.1216	0.02711	0.3234	37.59
Exact					

APPENDIX A: COMPUTER PROGRAM

The FORTRAN computer program is composed of a main and 9 subprograms. The subprograms are:

INFOR
SOLTN
VMATRX
ZMATRX
FIELD
ELSV
POTEN
INTG
GRAD

The last three programs compute the potential and its gradient at a point (x,y) due to the impressed charge distribution. Hence, as the source is charged, these programs must be changed accordingly.

<u>The Main Program:</u>

The main program reads in:

a) the number (NTOTAL) of the straight line segments approximately the boundary C of the region R.

b) the dielectric constant (SPSR) of the medium R.

c) the parameter LAPOIS. If LAPOIS is equal to zero, the problem is to solve the Laplace's equation. (In this case the last three subroutines are not needed). If LAPOIS is equal to 1, we are solving Poisson's equation and hence the potential and its gradient produced by the impressed sources must be provided by the last three subroutines.

For each of NTOTAL linear segments, the main program calls the subprogram INFOR. Then it calls the subroutine SOLTN.

<u>The INFOR subprogram:</u>

The subroutine INFOR reads in

a) The coordinates (X1, Y1) and X2, Y2) of the starting and ending points of each linear segment approximating the boundary C. (X1, Y1, X2, Y2 are in micrometers).

b) The number NSEC, of subsections that each linear segment is to be divided into.

c) For each linear segment α , β and σ are read in. These are sent back to the main program, where they are stored in the matrix BCOND.

In the subroutine INFOR, the coordinates of the starting and ending points of each subsection is computed. This information is stored in arrays XV1, YV1, XV2 and

<u>The subroutine SOLTN:</u>

In this subroutine the moment matrix equation is formed and solved. This subroutine calls various subroutines.

i) <u>The subroutine VMATRX:</u>

In this subroutine the right hand side of eqn. (7) is computed at the center of each subsection. The result is stored in the array V.

ii) <u>The subroutine ZMATRX:</u>

In this subroutine the moment matrix Z is computed. The (i,j) th element of this matrix is the right hand side of eqn. (7), computed at the center of jth subsection. (0 here is the potential produced by a constant charge density of $1(C/m^2)$ *on the jth subsection*).

iii) <u>The subroutine ELSV:</u>

This subroutine takes the inverse of the moment matrix Z and stores the inverse matrix into the Z matrix.

Once Z matrix is inverted, the surface density is computed in SOLTN subroutine by multiplying the inverse of the Z matrix by the column vector V. The charge density is stored in the array I.

iv) <u>The subroutine FIELD:</u>

This subroutine computes the total potential at K points equally spaced between the points. (XIN, YIN) and XFIN, YFIN)

The last three subroutines compute the potential and its gradient at a given point due to a constant volume charge density RHO (C/m^3) in a rectangular cylinder of infinite length.

INPUT/OUTPUT OF THE PROGRAM

The input to the program is through the data file 92. The first line of the input file is

NTOTAL, EPSE, LAPOIS

Then we have NTOTAL pairs of lines which have the form
X1, Y1, X2, Y2, NSEC

ALPHA, BETA, GAMA.

Where (X1, Y1) and (X2, Y2) denote the coordinates of the starting and ending point of a linear segment, and NSEC is the number of subsection, that the segment will be subdivided into ALPHA, BETA AND GAMA show the values of α , β and σ on the segment.

The last line in the input file has the form XIN, YIN, XFIN, YFIN, K where (X IN, YIN) and (XFIN, YFIN) are the coordinates of two points, and K is an integer. The program will compute the total potential at K equidistant points lying between the points (XIN, YIN) and (XFIN, YFIN).

The output of the program is printed in data file 18. Here the potential at K points is printed. X and Y are the coordinates of the point at which the potential is computed.

PROGRAM LISTING

The following is the listing of the program:

```
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C THIS PROGRAM COMPUTES THE  EQUIVALENT ELECTRIC              C
C CHARGE DENSITY ON THE SURFACE OF A LOSSLESS DIELECTRIC      C
C CYLINDER. THIS IS A TWO-DIMENSIONAL PROBLEM. THE TOTAL      C
C POTENTIAL INSIDE THE DIELECTRIC IS DUE TO SOME SPECIFIED    C
C CHARGES INSIDE AND DUE TO SOME IMPRESSED POTENTIALS         C
C ALONG THE SURFACE.                                          C
C                                                             C
C AT ANY POINT ON THE BOUNDARY OF THE CYLINDER WE HAVE        C
C                                                             C
C ALPA(C)*POT(CHARGE)+BETA(C)*(D/DN)(POT(CHARGE)) = -ALPHA(C) C
C *POT(SOURCE)-BETA(C)*(D/DN)POT(SOURCE)+GAMA(C)              C
C                                                             C
C WHERE;                                                      C
C C SHOWS THE VARIABLE ALONG THE BOUNDARY OF THE CYLINDER,    C
C ALPHA(C), BETA(C) AND GAMA(C) ARE THREE FUNCTIONS THAT ARE  C
C SPECIFIED AT ANY POINT C,                                   C
C POT(CHARGE)=POTENTIAL PRODUCED AT THE POINT C, BY THE       C
C UNKOWN EQUIVALENT SURFACE CHARGE RESIDING ON THE BOUNDARY   C
C OF THE CYLINDER,                                            C
C (D/DN) IS AN OPERATOR WHICH GIVES THE NORMAL DERIVATIVE     C
C OF THE FUNCTION THAT IT OPERATES ON , AND                   C
C POT(SOURCE) IS THE POTENTIAL PRODUCED BY THE IMPRESSED      C
C  SOURCES AT THE POINT ON THE BOUNDARY. THE IMPRESSED        C
C SOURCES ARE THE VOLUME CHARGE DENSITY INSIDE THE CYLINDER.  C
C THESE ARE THE SOURCES THAT APPEAR ON THE RIGHT              C
C HAND SIDE OF THE POISSON'S EQUATION.                        C
C                                                             C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C
      IMPLICIT COMPLEX*16 (C)
      IMPLICIT REAL*8(A-B,E-H,P-Z)
      DIMENSION  V(200),RI(200),Z(200,200)
      DIMENSION XV1(200),YV1(200),XV2(200),YV2(200)
      COMMON/NNN/NDP(10),BCOND(10,3),NTOTAL
      COMMON/TYPE/LAPOIS,EPSR
      NMAX=200
C
C  READ TOTAL NUMBER  ,NTOTAL,OF LINEAR SEGMENTS WHICH
C  CONSTITUTE THE BOUNDARY FOR THE PROBLEM. ALSO
C  READ THE DIELECTRIC CONSTANT ,EPSR, OF
C  THE MEDIUM.
C
C  IF LAPOIS IS ZERO THEN WE ARE SOLVING LAPLACE EQUATION
C  IF LAPOIS IS ONE THEN THE PROBLEM IS POISSON TYPE.
C
      READ(92,*) NTOTAL,EPSR,LAPOIS
      IF(LAPOIS.EQ.1)WRITE(93,1232)
      IF(LAPOIS.EQ.0)WRITE(93,1233)
 1232 FORMAT(/,25X,'THIS IS POISSON S EQUATION:',/)
 1233 FORMAT(/,25X,'THIS IS LAPLACE S EQUATION:',/)
      WRITE(93,1234)NTOTAL,EPSR
 1234 FORMAT(25X,'-----------------------------------'
     4 ,//,10X,'NO. OF TOTAL LINEAR SEGMENT BOUNDARIES=',I2,/,10X,
     4 'THE DIELECTRIC CONSTANT OF THE CYLINDER IS=',F8.4,//)
      NAI=0
      NBI=0
```

```
C
      DO 199 I=1,NTOTAL
C
C FOR EACH OF NTOTAL LINEAR SEGMENTS FORMING THE BOUNDARY
C CALL THE INFORMATION (INFOR) SUBROUTINE TO  ;
C A) READ IN THE COORDINATES (X1,Y1) OF THE INITIAL POINT AND (X2,Y2)
C    OF THE FINAL POINT OF I'TH LINEAR SEGMENT.
C B) READ IN THE NUMBER ,NSEC, OF SMALLER SUBSECTIONS THAT THIS PARTICULAR
C    LINEAR SEGMENT IS TO BE DIVIDED. THE CENTER OF EACH OF THESE
C    SUBSECTIONS IS A MATCHING POINT.
C C) READ THE VALUES ALPHA, BETA, AND GAMA FOR THIS PARTICULAR
C    LINEAR SEGMENT.
C D) FIND THE COORDINATES OF THE STARTING AND ENDING POINTS OF THESE
C     SUBSECTIONS AND STORE THEM IN THE ARRAYS XV1,YV1,XV2,YV2.
C
C
C
      CALL INFOR(XV1,YV1,XV2,YV2,NAI,NMAX,A,B,G)
      NDP(I)=NAI-NBI
      BCOND(I,1)=A
      BCOND(I,2)=B
      BCOND(I,3)=G
C
C WRITE THE BOUNDARY CONDITIONS DATA FOR THIS LINEAR SEGMENT;
C
      WRITE(93,111)
 111  FORMAT('1')
      WRITE(93,112) I,A,B,G
 112  FORMAT(////5X,'THIS IS THE INFORMATION OF THE
     $ BOUNDARY =',1X,I3,//,5X,'HERE ALPHA=',F9.5,3X,
     $ 'BETA=',E11.4,3X,'GAMA=',F9.5,/)
C
C WRITE GEOMETRICAL DATA FOR THIS LINEAR SEGMENT;
C
      WRITE(93,114)
 114  FORMAT(////12X,2('X-COORDINATE',5X,'Y-COORDINATE',5X))
      WRITE(93,115)(J,XV1(J),YV1(J),XV2(J),YV2(J),J=NBI+1,NAI)
 115  FORMAT(//(5X,I3,4(2X,1E))/)
      NBI=NAI
 199  CONTINUE
C
C OBTAIN THE TOTAL NUMBER OF UNKNOWNS IN THE MATRIX EQUATION.
C
      NUNKNS=NAI
      WRITE(93,993)NUNKNS
993   FORMAT(5X,'TOTAL NO. OF UNKNOWNS=',I3,//)
C
C CALL THE SOLUTION SUBROUTINE TO SOLVE THE PROBLEM.
C
      CALL SOLTN(Z,V,PI,XV1,YV1,XV2,YV2,NUNKNS,NMAX)
998   CONTINUE
      STOP
      END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      SUBROUTINE INFOR(XV1,YV1,XV2,YV2,NAI,NM,A,B,G)
      IMPLICIT COMPLEX*16 (C)
      IMPLICIT REAL*8(A-B,E-H,P-Z)
C
C IN THIS SUBROUTINE THE DATA IS ARRANGED IN THE PROPER FORM
```

```
C FOR FURTHER COMPUTATIONS.
C
         DIMENSION XV1(NM),YV1(NM),XV2(NM),YV2(NM)
         NNODES=NAI
C READ THE COORDINATES (X1,Y1) AND (X2,Y2),
C READ THE NUMBER OF SECTIONS OF THE BOUNDARY (NSEC),
C ALSO READ THE BOUNDARY CONDITIONS INFORMATION; ALPHA(A),
C BETA(B) AND GAMA(G)
C
C
         READ(92,*)X1,Y1,X2,Y2,NSEC
         X1=X1*1.D-06
         Y1=Y1*1.D-06
         X2=X2*1.D-06
         Y2=Y2*1.D-06
         READ(92,*)A,B,G
         EDELX=(X2-X1)/FLOAT(NSEC)
         EDELY=(Y2-Y1)/FLOAT(NSEC)
         DO 20 J=1,NSEC
         NNODES=NNODES+1
         XV1(NNODES)=X1+FLOAT(J-1)*EDELX
         YV1(NNODES)=Y1+FLOAT(J-1)*EDELY
 20      CONTINUE
         DO 70 I=NAI+1,NNODES-1
         XV2(I)=XV1(I+1)
         YV2(I)=YV1(I+1)
 70      CONTINUE
 75      XV2(NNODES)=X2
         YV2(NNODES)=Y2
 76      NAI=NNODES
         RETURN
         END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
       SUBROUTINE SOLTN(Z,V,RI,XV1,YV1,XV2,YV2,N,NM)
C
C IN THIS SUBROUTINE THE MATRIX EQUATION AX=Y IS SOLVED USING THE
C METHOD OF MOMENTS.
C
       IMPLICIT COMPLEX*16 (C)
         IMPLICIT REAL*8(A-B,E-H,P-Z)
       DIMENSION V(N),RI(N),Z(N,N),AUX1(600),AUX2(600)
       DIMENSION XV1(NM),YV1(NM),XV2(NM),YV2(NM)
       COMMON/NNN/NDP(10),BCOND(10,3),NTOTAL
C
C INTIALIZE THE VECTORS Z,V,AND RI.
C
       DO 5 I=1,N
       V(I)=0.D0
       RI(I)=0.D0
 5     CONTINUE
       DO 10 I=1,N
       DO 10 J=1,N
       Z(I,J)=0.D0
 10    CONTINUE
C
C CALL THE SUBROUTINE VMATRX TO COMPUTE THE EXCITATION VECTOR.
C
       CALL VMATRX(V,N,XV1,YV1,XV2,YV2,NM)
C
```

```
C CALL THE ZMATRX SUBROUTINE TO OBTAIN THE IMPEDANCE MATRIX.
C
      CALL ZMATRX(Z,XV1,YV1,XV2,YV2,N,NM)
C
C CALL THE ELSV SUBROUTINE TO INVERT THE MATRIX.
C
      EP=0.1D-09
      CALL ELSV(Z,AUX1,AUX2,N,DE,EP)
      WRITE(93,118)DE
 118  FORMAT(5X,'DE =',1E)
C
C MULTIPLY THE INVERSE OF Z-MATRIX WITH THE EXCITATION VECTOR
C TO OBTAIN THE CHARGES.
C
      DO 25 I=1,N
      SUM=0.D0
      DO 24 J=1,N
      SUM=SUM+Z(I,J)*V(J)
 24   CONTINUE
      RI(I)=SUM
 25   CONTINUE
C
C WRITE THE CHARGES ON THE OUTPUT FILE.
C
      NF=0
      DO 135 JKLM=1,NTOTAL
      NI=NF+1
      NF=NF+NDP(JKLM)
      WRITE(93,101)
 101  FORMAT('1')
      WRITE(93,102) JKLM
 102  FORMAT(//5X,'CHARGES ON THE  BOUND. =',1X,I2,//)
        ALPHA=BCOND(JKLM,1)
        BETA=BCOND(JKLM,2)
        GAMA=BCOND(JKLM,3)
        WRITE(93,198)ALPHA,BETA,GAMA
198     FORMAT(5X,'FOR THIS BOUNDARY    ALPHA=',E11.4,
     &3X,'BETA=',E11.4,3X,'GAMA=',E11.4,//)
      DO 30 I=NI,NF
      WRITE(93,105) I,RI(I)
 105  FORMAT(/1X,I3,5X,E11.4)
 30   CONTINUE
135     CONTINUE
        CALL FIELD(RI,N,XV1,XV2,YV1,YV2,NM)
        RETURN
      END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      SUBROUTINE VMATRX(V,N,XV1,YV1,XV2,YV2,NM)
C
C IN THIS SUBROUTINE THE EXCITATION VECTOR V IS
C COMPUTED .
C
      IMPLICIT COMPLEX*16 (C)
        IMPLICIT REAL*8(A-B,E-H,P-Z)
      DIMENSION V(N),XV1(NM),YV1(NM),XV2(NM),YV2(NM)
        COMMON/NNN/NDP(10),BCOND(10,3),NTOTAL
        COMMON/TYPE/LAPOIS,EPSR
        PI=4.D0*DATAN(1.D0)
        TP=2.D0*PI
        EPS=EPSR*8.854D-12
```

```
      IF(LAPOIS.EQ.1)TP=TP*EPS
      NF=0
      DO 110 JKLMN=1,NTOTAL
      NI=NF+1
      NF=NF+NDP(JKLMN)
      ALPHA=BCOND(JKLMN,1)
      BETA=BCOND(JKLMN,2)
      GAMA=BCOND(JKLMN,3)
      DO 100 I=NI,NF
      V(I)=GAMA*TP
      IF(LAPOIS.EQ.0)GO TO 100
      X1=XV1(I)
      Y1=YV1(I)
      X2=XV2(I)
      Y2=YV2(I)
      XF=(X1+X2)/2.0
      YF=(Y1+Y2)/2.0
      IF(ALPHA.EQ.0.0)GO TO 90
      CALL POTEN(XF,YF,POT)
      V(I)=V(I)-ALPHA*POT
C     IF(LAPOIS.EQ.1.AND.BETA.EQ.0.0)V(I)=10.0*V(I)
      IF(BETA.EQ.0.0)GO TO 100
 90   CALL GRAD(XF,YF,POTX,POTY)
      FNRMD=-(X2-X1)*POTY
      FNRMD=FNRMD+(Y2-Y1)*POTX
      FL=SQRT((X2-X1)*(X2-X1)+(Y2-Y1)*(Y2-Y1))
      V(I)=V(I)-BETA*FNRMD/FL
  100 CONTINUE
 110  CONTINUE
      RETURN
      END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      SUBROUTINE ZMATRX(Z,XV1,YV1,XV2,YV2,N,NM)
C
C IN THIS SUBROUTINE THE Z-MATRIX IS FORMED.
C
      IMPLICIT COMPLEX*16 (C)
        IMPLICIT REAL*8(A-B,E-H,P-Z)
      DIMENSION XV1(NM),YV1(NM),XV2(NM),YV2(NM),Z(N,N)
      COMMON/NNN/NDP(10),BCOND(10,3),NTOTAL
        COMMON/TYPE/LAPOIS,EPSR
      C1=DCMPLX(1.D0,0.D0)
      CK=DCMPLX(100.D0,0.D0)
      PI=4.D0*DATAN(1.D0)
        NF=0
        DO 1000 JKLM=1,NTOTAL
        ALPHA=BCOND(JKLM,1)
        BETA=BCOND(JKLM,2)
        NI=NF+1
        NF=NF+NDP(JKLM)
      DO 999 I=NI,NF
C
C COMPUTE THE PARAMETERS OF THE FIELD SUBSECTION
C
        XI=XV1(I)
        XIP1=XV2(I)
        YI=YV1(I)
        YIP1=YV2(I)
        CZI=CMPLX(XI,YI)
        CZIP1=DCMPLX(XIP1,YIP1)
```

```
      EDELI=CDABS(CZIP1-CZI)
      CUI=(CZIP1-CZI)/EDELI
      CZICAR=(CZI+CZIP1)/2.D0
      DO 888 J=1,N
C
C COMPUTE THE PARAMETERS OF THE SOURCE SUBSECTION
C
      XJ=XV1(J)
      XJP1=XV2(J)
      YJ=YV1(J)
      YJP1=YV2(J)
      CZJ=DCMPLX(XJ,YJ)
      CZJP1=DCMPLX(XJP1,YJP1)
      EDELJ=CDABS(CZJP1-CZJ)
      CUJ=(CZJP1-CZJ)/EDELJ
      CARG=(CZICAR-CZJP1)/(CZICAR-CZJ)
C
      CTERM=CDLOG(CARG)/CUJ
      IF(BETA.EQ.0.D0)GO TO 666
      IF(I.EQ.J)EDERV=PI
      IF(I.NE.J)EDERV=DIMAG(CUI*CTERM)
      Z(I,J)=BETA*EDERV
      IF(ALPHA.EQ.0.D0)GO TO 888
CCCCCCCCCCCCC
666   CT1=(CZICAR-CZJ)*CTERM
      CT2=EDELJ*(1.D0+CDLOG(CK/(CZICAR-CZJP1)))
      Z(I,J)=Z(I,J)+ALPHA*DREAL(CT1+CT2)
C     IF(LAPOIS.EQ.1.AND.BETA.EQ.0.D0)Z(I,J)=10.0*Z(I,J)
888   CONTINUE
999   CONTINUE
1000  CONTINUE
      RETURN
      END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      SUBROUTINE FIELD(RI,N,XV1,XV2,YV1,YV2,NM)
      IMPLICIT COMPLEX*16 (C)
      IMPLICIT REAL*8(A-B,E-H,P-Z)
      DIMENSION RI(N),XV1(NM),XV2(NM),YV1(NM),YV2(NM)
      COMMON/TYPE/LAPOIS,EPSR
      TPI=8.D0*DATAN(1.D0)
      IF(LAPOIS.EQ.1)TPI=TPI*EPSR*8.854D-12
      C1=(1.D0,0.D0)
      CK=(100.D0,0.D0)
C
C READ K=NO. OF POINTS AT WHICH FIELD AND POTENTIAL IS TO BE COMPUTED
C
      READ(92,*)XIN,YIN,XFIN,YFIN, K
      IF(K.EQ.1)IJKL=1
      IF(K.EQ.1)GO TO 78
      XDEL=(XFIN-XIN)/FLOAT(K-1)
      YDEL=(YFIN-YIN)/FLOAT(K-1)
 78   CONTINUE
      DO 50 J=1,K
      X=(XIN+FLOAT(J-1)*XDEL)*1.D-06
      Y=(YIN+FLOAT(J-1)*YDEL)*1.D-06
      IF(K.EQ.1.AND.J.NE.1)GO TO 50
      CZK=DCMPLX(X,Y)
      SUMP=0.D0
      SUMX=0.D0
      SUMY=0.D0
```

```
      DO 45 I=1,N
      CZI=DCMPLX(XV1(I),YV1(I))
      CZIP1=DCMPLX(XV2(I),YV2(I))
      EDELI=CDABS(CZIP1-CZI)
      CUI=(CZIP1-CZI)/EDELI
      CARG=(CZK-CZIP1)/(CZK-CZI)
      CTERM0=CDLOG(CARG)/CUI
      CTERM=(CZK-CZI)*CTERM0
      CTERM2=EDELI*(C1+CDLOG(CK/(CZK-CZIP1)))
      CWI=CTERM+CTERM2
      SUMP=SUMP+RI(I)*DREAL(CWI)
      SUMX=SUMX+RI(I)*DREAL(CTERM0)
      SUMY=SUMY-RI(I)*DIMAG(CTERM0)
      SUMF=SQRT(SUMX*SUMX+SUMY*SUMY)
 45   CONTINUE
      SUMP=SUMP/TPI
      SUMX=SUMX/TPI
      SUMY=SUMY/TPI
      SUMF=SUMF/TPI
C     WRITE(18,55)CZK,SUMP,SUMX,SUMY,SUMF
      IF(LAPOIS.EQ.0)GO TO 50
      CALL POTEN(X,Y,SP)
      SUMP=SUMP+SP/TPI
      WRITE(18,49)X,Y,SUMP
 49   FORMAT(3X,'X=',E11.5,2X,'Y=',E11.5,3X,'TOTAL POT=',E11.4,/)
 50   CONTINUE
 55   FORMAT(1X,'Z=',2E11.5,3X,'POT=',E10.4,3X,'EX=',E10.4,2X,
     6 'EY=',E10.4,3X,'ETOT=',E10.4)
      RETURN
      END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      SUBROUTINE ELSV(A,B,C,N,DE,EP)
      IMPLICIT REAL*8(A-H,P-Z)
      DIMENSION A(N,N),B(N),C(N)
      DO 11 I=1,N
      B(I)=0.D0
      C(I)=0.D0
      DO 12 J=1,N
 12   C(I)=C(I)+A(I,J)
 11   A(I,I)=A(I,I)-1.D0
      DO 13 K=1,N
      DO 14 J=1,N
      B(J)=A(K,J)
 14   A(K,J)=0.D0
      A(K,K)=1.D0
      W=B(K)+1.D0
      IF(ABS(W).LT.EP)GO TO 17
      DO 13 I=1,N
      Y=A(I,K)/W
      DO 13 J=1,N
 13   A(I,J)=A(I,J)-B(J)*Y
      DE=0.D0
      DO 15 J=1,N
      B(J)=0.D0
      DO 16 I=1,N
 16   B(J)=B(J)+A(I,J)
 15   DE=DE+C(J)*B(J)
      RETURN
 17   DE=-1.D0
      RETURN
```

END

```
      SUBROUTINE POTEN(X,Y,POT)
      IMPLICIT COMPLEX*16 (C)
      IMPLICIT REAL*8 (A-B,E-H,P-Z)
C
C
C THIS PROGRAM GIVES THE POTENTIAL AND THE
C GRADIENT OF THE POTENTIAL PRODUCED BY A
C CERTAIN TWO DIMENSIONAL CHARGE DISTRIBUTION
C AT A POINT (X,Y).
C
C NOTE THAT AS THE CHARGE DISTRIBUTION IS CHANGED
C THIS PROGRAM SHOULD BE CHANGED ACCORDINGLY.
C
C THIS PROGRAM WILL BE CALLED BY THE PROGRAM NAMED
C 'DNEWPOISSON' ONLY IF THE PARAMETER LAPOIS IS 1 IN
C THAT PROGRAM.
C
      RHO=-3200.D0
      AK=100.D0
      X1=-0.50D-06
      Y1=-0.04D-06
      X2=0.5D-06
      Y2=0.04D-06
CCCCCCCCCCCCCCCCCCC
      POT=(Y2-Y1)*(X2-X1)*DLOG(AK)
CCCCCCCCCCCCCCCCCCC
      EDEL1=DABS(X2-X1)
      U1=(X2-X1)/EDEL1
CCCCCCCCCCCCCCCCCCC
      FI1=(Y2-Y1)
      POT=POT+EDEL1*FI1
CCCCCCCCCCCCCCCCCCC
      CZ3=DCMPLX(X2,Y1)
      CZ4=DCMPLX(X2,Y2)
      CZ=DCMPLX(X,Y)
      EDEL2=CDABS(CZ4-CZ3)
      CU2=(CZ4-CZ3)/EDEL2
      CT2=(CZ-CZ3)*CDLOG((CZ-CZ4)/(CZ-CZ3))/CU2
      CT2=CT2+EDEL2*(1.D0-CDLOG(CZ-CZ4))
CCCCCCCCCCCCCCCCCCCCC
      FI2=DREAL(CT2)
      FI31=(X1-X)*FI2
      POT=POT+EDEL1*FI2+FI31/U1
CCCCCCCCCCCCCCCCCCCC
      XS=X2
      CALL INTG(XS,Y1,Y2,X,Y,RES32)
CCCCCCCCCCCCCCCCCCCCCC
      FI32=RES32
      POT=POT+FI32/U1
CCCCCCCCCCCCCCCCCCCC
      CZ5=DCMPLX(X1,Y1)
      CZ6=DCMPLX(X1,Y2)
      EDEL3=CDABS(CZ6-CZ5)
      CU3=(CZ6-CZ5)/EDEL3
      CT41=(CZ-CZ5)*CDLOG((CZ-CZ6)/(CZ-CZ5))/CU3
      CT41=CT41+EDEL3*(1.D0-CDLOG(CZ-CZ6))
CCCCCCCCCCCCCCCCCCCCCCCC
      FI41=(X1-X)*DREAL(CT41)
      POT=POT-FI41/U1
CCCCCCCCCCCCCCCCCCCCCCC
```

```
      XS=X1
      CALL INTG(XS,Y1,Y2,X,Y,RES42)
CCCCCCCCCCCCCCCCCCCCCCCCCCCC
      FI42=RES42
      POT=POT-FI42/U1
      POT=RHO*POT
      RETURN
      END
      SUBROUTINE INTG(XS,Y1,Y2,X,Y,RES)
      IMPLICIT REAL*8(A-B,E-H,P-Z)
      PI=4.D0*DATAN(1.D0)
      TP=2.D0*PI
      PO2=PI/2.D0
      PO4=PI/4.D0
      IF(X-XS)20,10,15
 10   IF(Y.GE.Y2)RES=PO4*((Y2-Y)*(Y2-Y)-(Y1-Y)*(Y1-Y))
      IF(Y.LE.Y1)RES=-PO4*((Y2-Y)*(Y2-Y)-(Y1-Y)*(Y1-Y))
      IF(Y.GT.Y1.AND.Y.LT.Y2)IM=1
      IF(IM.EQ.1)RES=-PO4*((Y2-Y)*(Y2-Y)+(Y1-Y)*(Y1-Y))
      RETURN
 15   T1=(X-XS)*(Y2-Y1)
      T2=(Y-Y2)*(Y-Y2)+(X-XS)*(X-XS)
      T3=(Y-Y1)*(Y-Y1)+(X-XS)*(X-XS)
      T4=(Y2-Y)/(X-XS)
      T5=(Y1-Y)/(X-XS)
      IF(Y.GE.Y2)RES=(T1-T2*DATAN(T4)+T3*DATAN(T5))/2.D0
      IF(Y.LE.Y1)RES=(T1-T2*DATAN(T4)+T3*DATAN(T5))/2.D0
      IF(Y.GT.Y1.AND.Y.LT.Y2)IM=1
      IF(IM.EQ.1)RES=(T1+T3*DATAN(T5)-T2*DATAN(T4))/2.D0
      RETURN
 20   T1=(XS-X)*(Y2-Y1)
      T2=(Y2-Y)*(Y2-Y)+(XS-X)*(XS-X)
      T3=(Y-Y1)*(Y-Y1)+(XS-X)*(XS-X)
      T4=(Y2-Y)/(XS-X)
      T5=(Y1-Y)/(XS-X)
      TERM=PO2*((Y2-Y)*(Y2-Y)-(Y1-Y)*(Y1-Y))
      IF(Y.GE.Y2)RES=TERM-(T1-T2*DATAN(T4)+T3*DATAN(T5))/2.D0
      IF(Y.LE.Y1)RES=-TERM+(T2*DATAN(T4)-T1-T3*DATAN(T5))/2.D0
      IF(Y.LT.Y2.AND.Y.GT.Y1)IM=1
      IF(IM.EQ.1)TM=-PO2*((Y1-Y)*(Y1-Y)+(Y2-Y)*(Y2-Y))
      IF(IM.EQ.1)RES=TM-(T1+T3*DATAN(T5)-T2*DATAN(T4))/2.D0
      RETURN
      END
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
      SUBROUTINE GRAD(X,Y,GX,GY)
      implicit complex*16 (c)
      IMPLICIT REAL*8(A-B,E-H,P-Z)
      RHO=-3200.D0
      CZ=DCMPLX(X,Y)
      X1=-0.50D-06
      Y1=-0.04D-06
      X2=0.5D-06
      Y2=0.04D-06
      CZ1=DCMPLX(X1,Y1)
      CZ2=DCMPLX(X2,Y1)
      CZ3=DCMPLX(X1,Y2)
      CZ4=DCMPLX(X2,Y2)
      EDEL1=CDABS(CZ2-CZ1)
      CU1=(CZ2-CZ1)/EDEL1
      EDEL2=CDABS(CZ4-CZ3)
```

```
CU2=(CZ4-CZ3)/EDEL2
CTERM1=(CZ-CZ1)*CDLOG((CZ-CZ2)/(CZ-CZ1))/CU1
CTERM1=CTERM1+EDEL1*(1.D0+CDLOG(1.D0/(CZ-CZ2)))
CTERM2=(CZ-CZ3)*CDLOG((CZ-CZ4)/(CZ-CZ3))/CU2
CTERM2=CTERM2+EDEL2*(1.D0+CDLOG(1.D0/(CZ-CZ4)))
GY=-RHO*DREAL(CTERM2-CTERM1)
GX=0.D0
RETURN
END
```

SAMPLE INPUT/OUTPUT FILE:

The following is the input/output file for the long channel MOSFET problem considered in Fig. 4. The results presented in the following are plotted in Fig. 6 and 7.

```
4  11.750 1
0.5 -0.106  0.5 0.106  12
1.0 0.0 5.9
-0.5 0.106 -0.5 -0.106 12
1.0 0.0 0.9
-0.5 -0.106 0.5 -0.106 25
0.0 1.0 0.0
0.5  0.106  -0.5  0.106  25
1.0 0.22E-06  2.0
-0.50 0.1059999 0.50 0.105999 21
```

```
X=-.50000E-06  Y=0.10600E-06   TOTAL POT= 0.9552E+00
X=-.45000E-06  Y=0.10600E-06   TOTAL POT= 0.1016E+01
X=-.40000E-06  Y=0.10600E-06   TOTAL POT= 0.1005E+01
X=-.35000E-06  Y=0.10600E-06   TOTAL POT= 0.9841E+00
X=-.30000E-06  Y=0.10600E-06   TOTAL POT= 0.9704E+00
X=-.25000E-06  Y=0.10600E-06   TOTAL POT= 0.9691E+00
X=-.20000E-06  Y=0.10600E-06   TOTAL POT= 0.9827E+00
X=-.15000E-06  Y=0.10600E-06   TOTAL POT= 0.1012E+01
X=-.10000E-06  Y=0.10600E-06   TOTAL POT= 0.1059E+01
X=-.50000E-07  Y=0.10600E-06   TOTAL POT= 0.1126E+01
X=0.00000E+00  Y=0.10600E-06   TOTAL POT= 0.1216E+01
X=0.50000E-07  Y=0.10600E-06   TOTAL POT= 0.1332E+01
X=0.10000E-06  Y=0.10600E-06   TOTAL POT= 0.1478E+01
X=0.15000E-06  Y=0.10600E-06   TOTAL POT= 0.1661E+01
X=0.20000E-06  Y=0.10600E-06   TOTAL POT= 0.1891E+01
X=0.25000E-06  Y=0.10600E-06   TOTAL POT= 0.2178E+01
X=0.30000E-06  Y=0.10600E-06   TOTAL POT= 0.2534E+01
X=0.35000E-06  Y=0.10600E-06   TOTAL POT= 0.2986E+01
X=0.40000E-06  Y=0.10600E-06   TOTAL POT= 0.3575E+01
X=0.45000E-06  Y=0.10600E-06   TOTAL POT= 0.4387E+01
X=0.50000E-06  Y=0.10600E-06   TOTAL POT= 0.5556E+01
```

```
1  11.750 1
2.5 -0.106  2.5 0.106  12
1.0 0.0 5.9
-2.5 0.106 -2.5 -0.106 12
1.0 0.0 0.9
-2.5 -0.106 2.5 -0.106 50
0.0 1.0 0.0
2.5  0.106  -2.5  0.106  50
1.0 0.22E-06  2.0
-2.50 0.1059999 2.50 0.105999 21
```

```
X=-.25000E-05  Y=0.10600E-06   TOTAL POT= 0.9238E+00
X=-.22500E-05  Y=0.10600E-06   TOTAL POT= 0.7867E-00
X=-.20000E-05  Y=0.10600E-06   TOTAL POT= 0.6475E-00
X=-.17500E-05  Y=0.10600E-06   TOTAL POT= 0.5955E+00
X=-.15000E-05  Y=0.10600E-06   TOTAL POT= 0.5764E+00
X=-.12500E-05  Y=0.10600E-06   TOTAL POT= 0.5694E+00
X=-.10000E-05  Y=0.10600E-06   TOTAL POT= 0.5669E+00
X=-.75000E-06  Y=0.10600E-06   TOTAL POT= 0.5659E+00
X=-.50000E-06  Y=0.10600E-06   TOTAL POT= 0.5656E+00
X=-.25000E-06  Y=0.10600E-06   TOTAL POT= 0.5655E+00
X=0.00000E+00  Y=0.10600E-06   TOTAL POT= 0.5656E+00
X=0.25000E-06  Y=0.10600E-06   TOTAL POT= 0.5659E+00
X=0.50000E-06  Y=0.10600E-06   TOTAL POT= 0.5667E+00
X=0.75000E-06  Y=0.10600E-06   TOTAL POT= 0.5692E+00
X=0.10000E-05  Y=0.10600E-06   TOTAL POT= 0.5758E+00
X=0.12500E-05  Y=0.10600E-06   TOTAL POT= 0.5940E+00
X=0.15000E-05  Y=0.10600E-06   TOTAL POT= 0.6434E+00
X=0.17500E-05  Y=0.10600E-06   TOTAL POT= 0.7788E+00
X=0.20000E-05  Y=0.10600E-06   TOTAL POT= 0.1147E+01
X=0.22500E-05  Y=0.10600E-06   TOTAL POT= 0.2158E+01
X=0.25000E-05  Y=0.10600E-06   TOTAL POT= 0.5499E+01
```



Fig. 1. Geometry of the problem.



Fig. 2. The potential in R is produced by the impressed volume charges ρ_v and the equivalent surface charges σ. The surface charges are on C^+, (just outside of bounding surface C).



Fig. 3. A line charge at the center of an infinitely long grounded, rectangular pipe.



Fig. 4. An approximate model for a rectangular depletion region in a MOSFET.

($E_{si} = E_o E_r$, 9 Na = -3200.0 $C/3_m$)



Fig. 5. Rectangular Depletion Region under the gate of an N-Channel MOSFET



Fig. 6. Surface Potential Variation Along the Channel for a 1 μ m Device

______ Poole & Kwang (Ref. 4)
●...●...● Present Method



Fig. 7. Surface Potential Variation Along the Channel for a 5 μ m Device

Poole & Kwang (Ref. 4)



Fig. 8. Threshold Voltage Versus Channel Length for Drain Voltage (V_D) of 0 & 5V

——— Poole & Kwang (Ref. 4)

●...●...● Present Method

END
DTIC
8-86